QUERY Result and Meta Data

<cfquery 


name="qBlog" 


datasource="#REQUEST.DSN.Source#" 


username="#REQUEST.DSN.Username#" 


password="#REQUEST.DSN.Password#"


result="Query">


SELECT TOP 10



b.id,



b.name


FROM



blog_entry b


ORDER BY




b.id DESC

</cfquery>

<!--- Get the SQL and query request data. --->

<cfdump var="#Query#" />

<!--- Get the query Meta data. --->

<cfdump var="#GetMetaData( qBlog )#" />

Creating Queries Manually
When creating a query you can create an empty query, pass in just the column names, or pass in both the column names and column data types:

<!--- Create an empty query. --->

<cfset qGirl = QueryNew( "" ) />

<!--- Create a query with just column name. --->

<cfset qGirl = QueryNew( "name" ) />

<!--- Create a query with name and column type. --->

<cfset qGirl = QueryNew( "name", "CF_SQL_VARCHAR" ) />

When you do NOT pass in the query column type, ColdFusion tries to use the best guess. It will check your top 50 or 100 records to see what the lowest common denominator is and then use that type for the column.
Supported data types for QueryNew():

· Integer: 32-bit integer 

· BigInt: 64-bit integer 

· Double: 64-bit decimal number 

· Decimal: Variable length decimal, as specified by java.math.BigDecimal 

· VarChar: String 

· Binary: Byte array 

· Bit: Boolean (1=True, 0=False) 

· Time: Time 

· Date: Date (can include time information) 

You can pass in the types above, OR pass in the CF_SQL_ Query param style (ex. CF_SQL_INTEGER instead of Integer). The style does not make any difference.
Modifying The Query Structure

There are two primary ways to alter the structure of the query – adding rows and adding columns.

Adding Rows
<!--- Add a single row. --->

<cfset QueryAddRow( qGirl ) />

<!--- Add four rows. --->

<cfset QueryAddRow( qGirl, 4 ) />

When new rows get added, the cell values are NULL in the Java sense (ColdFusion doesn’t really like NULL values). Can query for these rows like they are NULL sql values (using query of queries):
<cfquery name="qNullGirl" dbtype="query">


SELECT * FROM qGirl WHERE name IS NULL
</cfquery>

<cfdump var="#qNullGirl#" />

Adding Columns
Adding columns is very exciting because it gives us the ability to define default columns. When adding a column, you must specify the query, the column name, and an array of default values. You can also set the column type (otherwise, CF will guess). The array does NOT need have data:
<!--- Add a column and specify default values. --->

<cfset QueryAddColumn( qGirl, "hotness", ArrayNew( 1 ) ) />

<!--- Add a column and specify type AND default values. --->

<cfset QueryAddColumn( qGirl, "hotness", "CF_SQL_VARCHAR", ArrayNew( 1 ) ) />

QueryAddColumn() is especially exciting because it gives us an easy way to build test queries:
<!--- Create an empty query. --->

<cfset qGirl = QueryNew( "" ) />

<!--- Populate the query. --->

<cfset QueryAddColumn(


qGirl, 


"name",


"CF_SQL_VARCHAR",


ListToArray(



"Sarah,Azure,Kit,Libby"



)


) />

<!--- Populate the query. --->

<cfset QueryAddColumn(


qGirl, 


"hotness",


"CF_SQL_DECIMAL",


ListToArray(



"8.5,9.0,7.5,9.0"



)


) />

Query of Queries
You can also alter an existing query using query of queries:

<!--- Add hair column using q-of-q. --->

<cfquery name="qGirl" dbtype="query">


SELECT



name,



hotness,



'' AS hair


FROM



qGirl

</cfquery>

Notice that our query name is the SAME as the query we are selecting from. This will overwrite the existing query with the new one.
Setting Query Values Manually
When setting query cell values, you can use either the dot (structure) notation or array notation:
<!--- Set using dot notation. --->

<cfset qGirl.hair = JavaCast( "string", "Brunette" ) />

<!--- Set using array notation. --->

<cfset qGirl[ "hair" ][ 1 ] = JavaCast( "string", "Red" ) />

If you use dot notation, it will set the value of the “current” row. If you use array notation, you can specify the row you want to set. NOTE: If you use array notation, you MUST use a row (ie. You can’t just use [ “hair” ]).
You can also use QuerySetCell():

<!--- Set using without row number - defaults to last record. --->

<cfset QuerySetCell( qGirl, "hair", "Red" ) />

<!--- Set using row number. --->

<cfset QuerySetCell( qGirl, "hair", "Red", qGirl.CurrentRow ) />

QuerySetCell() does the same exact thing as setting the query explicitly except it adds the overhead of a method call. However, it does default to last row, which some people like.
Data Type Casting
Notice that I have been JavaCast()’ing the values as I set them into the query cell. This is a MUST. ColdFusion does NOT have data types. Java DOES. In order to help ColdFusion make the leap, you have to tell it what data type to use. If you do not use this, further query-of-query calls may break depending on what actions you are performing (ex. ORDER BY). 
JavaCast() allows the following data types:

· boolean 

· int 

· long 

· float 

· double 

· string 

· null 

Be careful – these to no map quite nicely to the data types allowed in QueryNew() or QueryAddColumn(). You have to play around to figure out which data type maps to which data type (ex. CF_SQL_DECIMAL to float). 

Crazy Insane Usage

Here’s something the other tour guides won’t tell you… you can store just about anything you want into a query cell and then use it as an iterator:

<!--- Create the query. --->

<cfset qTest = QueryNew( "object" ) />

<!--- Add some rows. --->

<cfset QueryAddRow( qTest, 3 ) />

<!--- Store ColdFusion data types. --->

<cfset qTest[ "object" ][ 1 ] = ListToArray( "a,b,c" ) />

<cfset qTest[ "object" ][ 2 ] = StructNew() />

<cfset qTest[ "object" ][ 3 ] = ListToArray( "1,2,3" ) />

<cfloop query="qTest">


<cfdump var="#qTest.object#" />

</cfloop>
Of course, this is NOT *supported*. This is not a proper use and may not work in future versions of CF.
Referencing Query Values
When referring to a query cell value, you can use dot notation or array notation:
<!--- User dot notation. ---> 

#qGirl.name# 

<!--- User array notation. --->

#qGirl[ "name" ][ 1 ]#

When using array notation, you must specify a row number. Using the dot notation, you will reference the “current” row value.
Nested Queries
When using one query loop inside of another, be careful of how the references act:
<!--- Loop over both queries. --->

<cfloop query="qOuter">


<cfloop query="qInner">



#qOuter.value# - #qInner.value#<br />


</cfloop>

</cfloop>

… gives us:

A - 1
A - 2
A - 1
A – 2

Notice that the outer query reference always defaults to the first record. This is because from within the context of the inner loop, CF does not “know” where the current row pointer of the outer loop is – CF keeps a context so that can refer to cell values without a query scope – inner and outer conflict.
To over come this, use array notation:

<!--- Loop over both queries. --->

<cfloop query="qOuter">


<cfloop query="qInner">



#qOuter[ "value" ][ qOuter.CurrentRow ]# - #qInner.value#<br />


</cfloop>

</cfloop>
… gives us:

A - 1
A - 2
B - 1
B – 2
Notice that now, our output correctly refers to the intended variables.
Passing Queries

When you pass a query to a function, it is passed by reference. Since the query keeps track of its own internal row pointer, the passed query will know what row it is currently on, even if it is in a totally different context:
<cffunction name="GetCellValue">


<cfreturn ARGUMENTS[ 1 ].name />

</cffunction>

<cfloop query="qGirl">


#GetCellValue( qGirl )#<br />

</cfloop>

… this gives us:
Sarah
Azure
Kit
Libby
Notice that in the function body, we are returning the name value using dot notation. Remember, this grabs the value from the current row. Works quite nicely.
Misc. Query Stuff
Each query column can be treated like an array to some degree. You can run things like:

· ArraySum()

· ArrayAvg()

· ArrayLen()

· ArrayMax()

· ArrayMin()

However, when passing in the query column, you CANNOT use dot notation. You MUST use array notation:

<!--- Sum the value column. --->

<cfset sum = ArraySum( query[ “value” ] ) /> 

However, there are other, more “standard” query column functions:
· ValueList()

· QuotedValueList()

This CANNOT use array notation. You MUST use dot notation:

#ValueList( qGirl.name )#

Duplicating A Query
Queries can be duplicated using the Duplicate() method:
<!--- Duplicate the girl query. --->

<cfset qGirlCopy = Duplicate( qGirl ) />

This acts just as you would like it to.

Underlying Java Methods
Under the query object and the query column object, there are plenty of fun Java methods to use. I am not going to cover them here as Java-in-CF is a HUGE topic unto itself. For example, you can actually see if an “empty” field is truly an empty string or is actually a NULL java value.
Misc.
MaxRows attribute limits the number of rows that get returned in a query (query of queries as well).

Looping over a query as an INDEX array (using array notation) is faster than a query loop.
